<%
Class AssignAction
	Sub initialize
		
		'开启无缓存模式
		Call POP_MVC.NoCache
		
		'指定数据库路径为session中存储的assignedBook
		Call C_( Array("DB_PATH" , C_("db_dir") & S_("assignedBook")) )		
	End Sub


	'指定一个工作簿workBook来操作，该工作青筋表的名称将会被保存到session中
	sub assignBook
		
		'如果session中的工作簿名与修改值不同则重新赋值session
		if S_("assignedBook") <> that.get("book") Then			
			Call S_( array("assignedBook", that.get("book") ) )
			Call S_( array("assignedSheet", null ) )	'工作簿变了，也得修改当前指定的工作表
		end if
		if isEmpty(S_("assignedBook")) OR S_("assignedBook") = "" then
			call that.error( "指定失败" )
		else
			call that.success( "已经指定使用的工作簿为“" & that.get("book") & "”" )
		end if
	end sub
	
	'显示所有工作簿
	sub listBooks
		that.assign "list" , getBooks()
		that.display ""
	end sub
	
	'指定一个工作表workSheet来操作，该工作青筋表的名称将会被保存到session中
	sub assignSheet
		'工作表名后面默认带符号$，如“1学期$”，此时无法存进session，所以得先去掉$
		Call S_( array( "assignedSheet" , mid(that.get("sheet"),1,len(that.get("sheet"))-1 )) )
		if isEmpty(S_("assignedSheet")) OR "" =  S_("assignedSheet") then
			call that.error( "指定失败" )
		else
			call that.success( "已经指定使用的工作表为“" & S_("assignedSheet") & "”" )
		end if
	end sub
	
	'显示所有工作表
	sub listSheets
		that.assign "list",P_("excel").getTables().items
		
		that.display ""
	end sub
	
	'获取所有工作簿
	Function getBooks
		dim fd,fc,f1,arr
		set fd = POP_MVC.fso.getFolder( POP_MVC.realPath(APP_PATH & "/data/") )	'创建folder对象
		
		set fc = fd.files	'获取目录下的所有文件对象
		for each f1 in fc	'循环每个文件
			if POP_MVC.String.rstr(f1.name,".") = ".xls" Then	'如果文件名的后缀为.xls才存储
				POP_MVC.Arr.Push arr,f1.name
			end If
		next
		set fc = nothing
		set fd = nothing
		getBooks = arr
	end Function
End Class
%>